
Поточная (поочередная) печать ДВУХ отчетов с заданным количеством экземпляров для каждого



Решает примерно такую задачу:
Вывод на печать Счета-Фактуры и Акта выполненных работ для всех клиентов кои обслуживались в прошлом месяце, причем так что бы документы для каждого клиента лежали в лотке принтера рядом.

'==============================================================
'ПРИМЕР ИСПОЛЬЗОВАНИЯ
Private Sub TestProcedure()
'Поточная (поочередная) Счетов Фактур и Актов Выполненных Работ
'В следующем порядке:
' 2 экз. Счета Фактуры для одного клиента
' Потом 1 экз. Акта Выполненных Работ для него же
' дальше все тоже самое для сдедующего клиента
Call jsPipeLinePrint("Rp_СчетаФактуры", 2, "Rp_АктыВыполненныхРабот", 1)
End Sub
'==============================================================
Public Sub jsPipeLinePrint(ReportNameOne As String, CopiesOne As Byte, _
ReportNameTwo As String, CopiesTwo As Byte)
'js 12.09.2001
'Поточная (поочередная) печать ДВУХ отчетов с заданным количеством
'экземпляров для каждого
'Причем базовый набор записай для обоих отчетов один и тот-же
'либо они (наборы записей отчетов)полностью одинаковы по
'ко-ву записей и сортировке т.е. Имя Клиента на, допустим,
' 1й,10й,и 22й стр. одного отчета полностью совпадает с
' Именем Клиента на тех-же страницах другого

On Error GoTo jsPipeLinePrintErr

Dim TotPages As Integer 'Количество листов бумаги необходимых
'для печати всего обьема
Dim i As Integer 'Счетчик цикла
Dim x As Integer 'Концовка цикла

'Приостанавливаем вывод на экран (не обязательно)
Application.Echo False

'Открываем первый отчет Счета Фактуры
DoCmd.OpenReport ReportNameOne, acPreview
DoCmd.Minimize
'скрываем окно отчета (не обязательно)
Reports(ReportNameOne).Visible = False

'Открываем второй отчет Акты Выполненных Работ
DoCmd.OpenReport ReportNameTwo, acPreview
DoCmd.Minimize
'скрываем окно отчета (не обязательно)
Reports(ReportNameTwo).Visible = False

'Включаем вывод на экран
Application.Echo True

'Проверка на наличие данных в открытом отчете (только в одном)
'т.к. подразумевается что набор зап. второго полностью аналогичен
If Reports(ReportNameOne).HasData = 0 Then
MsgBox "Нечего печатать!!! " & vbCrLf & _
"отчеты:" & vbCrLf & _
"[" & ReportNameOne & "] и " & vbCrLf & _
"[" & ReportNameTwo & "]" & vbCrLf & _
"не имеют данных", vbCritical
'переход на метку закрытия отчетов
GoTo jsPipeLinePrintExit
End If

'Подсчет необходимых листов бумаги и установка концовки цикла
TotPages = Reports(ReportNameOne).Pages
x = TotPages
TotPages = TotPages * CopiesOne + TotPages * CopiesTwo

If MsgBox("Вставьте в принтер " & TotPages & " листов !", vbOKCancel) = vbOK Then
For i = 1 To x
DoCmd.SelectObject acReport, ReportNameOne
'печатаем со стр. i по стр. i с высоким разрешением заданное к-во копий
DoCmd.PrintOut acPages, i, i, , CopiesOne, False

DoCmd.SelectObject acReport, ReportNameTwo
'печатаем со стр. i по стр. i с высоким разрешением заданное к-во копий
DoCmd.PrintOut acPages, i, i, , CopiesTwo, False
Next i
End If

jsPipeLinePrintExit:
On Error Resume Next
DoCmd.Close acReport, ReportNameOne
DoCmd.Close acReport, ReportNameTwo
Exit Sub

jsPipeLinePrintErr:
Application.Echo True
MsgBox Err.Description & ""
Debug.Print Err.Description & vbCrLf & "Ощибка №" & Err.Number
Err.Clear
Resume jsPipeLinePrintExit
End Sub




